{$R+,G+,Q+}


PROGRAM TICTACTOE_MIT_ALPHA_BETA_und_Bewertungsliste;


 USES CRT;


 TYPE INHALT=RECORD


              ART,ANZAHL:BYTE;


              WO:ARRAY[1..4]OF BYTE


             END;


      BINHALT=RECORD WERT,ANZAHL:WORD END;


      NR=RECORD FELDNR:ARRAY[1..3]OF BYTE END;


      SPIELFELD=ARRAY[1..9]OF INHALT;


      BEWERTUNG=ARRAY[1..8]OF BINHALT;


      SPIELZUG=BYTE;


      SPIELZUGLISTE=ARRAY[1..9]OF SPIELZUG;


      SPIELER=(COMPUTER,MENSCH);


 CONST BELL=#7;CODEFARBE=MAGENTA;M_COLOR=LIGHTBLUE;


       C_COLOR=LIGHTRED;MININT=-30000;MAXINT=30000;


       COSUM:ARRAY[1..3]OF WORD=(5,100,5000);


       SPSUM:ARRAY[1..3]OF WORD=(10,200,10000);


 VAR REIHE:ARRAY[1..8]OF NR;


     F:SPIELFELD;


     B:BEWERTUNG;


     DRAN:SPIELER;


     TIEFE,ZUEGE:BYTE;


     COUNT:LONGINT;


     CH:CHAR;


     ANZEIGEN:BOOLEAN;





 PROCEDURE BLINKEN;


  BEGIN TEXTCOLOR(LIGHTCYAN+BLINK) END;


 PROCEDURE REVERSE;


  BEGIN TEXTCOLOR(GREEN+BLINK) END;


 PROCEDURE FETT;


  BEGIN TEXTCOLOR(MAGENTA) END;


 PROCEDURE UNTERSTREICHEN;


  BEGIN TEXTCOLOR(BLUE) END;


 PROCEDURE BLINK_LOW;


  BEGIN TEXTCOLOR(CYAN) END;


 PROCEDURE NORMAL;


  BEGIN TEXTCOLOR(YELLOW) END;


 PROCEDURE WARTE;


  VAR C:CHAR;


  BEGIN


   GOTOXY(45,23);WRITE('Drücke die Leertaste');


   REPEAT C:=READKEY UNTIL C=' ';


   GOTOXY(45,23);CLREOL


  END;





 PROCEDURE INIT;


  VAR K,L,I:BYTE;


  BEGIN


   FOR K:=1 TO 3 DO


    BEGIN


     FOR L:=1 TO 3 DO REIHE[L].FELDNR[K]:=(L-1)*3+K;


     FOR L:=4 TO 6 DO REIHE[L].FELDNR[K]:=3*K+L-6;


     REIHE[7].FELDNR[K]:=7-2*(K-1);


     REIHE[8].FELDNR[K]:=1+4*(K-1)


    END;


   WRITELN('Gewinnreihen');


   FOR L:=1 TO 8 DO


    BEGIN


     WRITE(L,') ');


     FOR K:=1 TO 3 DO WRITE(REIHE[L].FELDNR[K]:2);


     WRITELN


    END;


   FOR I:=1 TO 9 DO


    BEGIN


     F[I].ART:=0;F[I].ANZAHL:=0;


     FOR L:=1 TO 8 DO


      FOR K:=1 TO 3 DO


       IF REIHE[L].FELDNR[K]=I THEN


        BEGIN INC(F[I].ANZAHL);F[I].WO[F[I].ANZAHL]:=L END


    END;


   WRITELN;WRITELN('Feldinhalte');


   FOR I:=1 TO 9 DO


    BEGIN


     WRITE(I,') ');


     FOR K:=1 TO F[I].ANZAHL DO WRITE(F[I].WO[K]:2);


     WRITELN


    END;


   WARTE


  END;





 PROCEDURE NEUES_SPIEL(VAR AMZUG:SPIELER);


  BEGIN


   ZUEGE:=1;CLRSCR;NORMAL;


   GOTOXY(17,10);WRITE('Spielfeld');


   GOTOXY(15,11);WRITE('(((((((');


   GOTOXY(15,12);TEXTCOLOR(CODEFARBE);WRITE('1');NORMAL;WRITE('( ( ( ((');


   GOTOXY(15,13);TEXTCOLOR(CODEFARBE);WRITE('2');NORMAL;WRITE('( ( ( ((');


   GOTOXY(15,14);TEXTCOLOR(CODEFARBE);WRITE('3');NORMAL;WRITE('');


   GOTOXY(15,15);WRITE(' (((((((');


   GOTOXY(15,16);TEXTCOLOR(CODEFARBE);WRITE('   1  2  3');NORMAL;


   GOTOXY(50,1);WRITE('Tic Tac Toe  ');


   GOTOXY(35,3);WRITE('Ziel des Spiels ist, drei eigene Steine in');


   GOTOXY(35,4);WRITE('eine horizontale, diagonale oder vertikale');


   GOTOXY(35,5);WRITE('Reihe zu legen. ');


   GOTOXY(35,8);WRITE('Spielstärke (1-9) : ');


   REPEAT CH:=READKEY UNTIL CH IN['1'..'9'];WRITE(CH);


   TIEFE:=ORD(CH)-ORD('0');


   GOTOXY(35,17);WRITE('Wer soll beginnen (');TEXTCOLOR(WHITE);WRITE('M');


   NORMAL;WRITE(')ensch oder (');TEXTCOLOR(WHITE);WRITE('C');


   NORMAL;WRITE(')omputer ?');


   REPEAT CH:=UPCASE(READKEY) UNTIL CH IN[#13,'M','C'];


   IF CH IN[#13,'M'] THEN AMZUG:=MENSCH ELSE AMZUG:=COMPUTER;


   GOTOXY(35,17);CLREOL;


   WRITE('Bewertung anzeigen (J/N) : ');


   REPEAT CH:=UPCASE(READKEY) UNTIL CH IN['J','N'];


   WRITE(CH);DELAY(500);


   IF CH='J' THEN ANZEIGEN:=TRUE ELSE ANZEIGEN:=FALSE;


   GOTOXY(35,17);CLREOL;


  END;





 PROCEDURE MACHZUG(VAR F:SPIELFELD;VAR B:BEWERTUNG;ZUG:SPIELZUG;AMZUG:SPIELER);


  VAR L:BYTE;


  BEGIN


   IF AMZUG=COMPUTER THEN F[ZUG].ART:=1 ELSE F[ZUG].ART:=2;


   FOR L:=1 TO F[ZUG].ANZAHL DO


    WITH B[F[ZUG].WO[L]] DO


     BEGIN WERT:=WERT OR F[ZUG].ART;INC(ANZAHL) END;


  END;





 PROCEDURE SETZEN(ZUG:BYTE;STEIN:CHAR);


  VAR ZEILE,SPALTE,I:INTEGER;


  BEGIN


   ZEILE:=ZUG DIV 3+1;


   SPALTE:=ZUG MOD 3;


   IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(ZEILE) END;


   FOR I:=1 TO 5 DO


    BEGIN


     GOTOXY(SPALTE*3+15,11+ZEILE);WRITE(' ');DELAY(70);


     GOTOXY(SPALTE*3+15,11+ZEILE);WRITE(STEIN);DELAY(70)


    END


  END;





 PROCEDURE SPIELERZUG;


  VAR NOCHMAL:BOOLEAN;


      FELD,ZEILE,SPALTE:INTEGER;


FUNCTION EINGABE:INTEGER;


   VAR CH:CHAR; I:INTEGER;


   BEGIN


    REPEAT


     WRITE(BELL);


     REPEAT CH:=READKEY UNTIL CH IN ['1'..'3',#27];


     IF CH=#27 THEN BEGIN NOCHMAL:=TRUE;EINGABE:=1 END


     ELSE I:=ORD(CH)-ORD('0');


    UNTIL CH IN ['1'..'3',#27];


    IF NOT NOCHMAL THEN BEGIN WRITE(CH);EINGABE:=I END


   END;


  BEGIN


   REPEAT


    NOCHMAL:=FALSE;


    GOTOXY(43,15);WRITE('Bitte gib deinen Zug ein : ');CLREOL;


    GOTOXY(52,19);WRITE(' Zeile  : ');


    IF NOT NOCHMAL THEN ZEILE:=EINGABE


    ELSE BEGIN GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL END;


    GOTOXY(52,20);WRITE(' Spalte : ');


    IF NOT NOCHMAL THEN SPALTE:=EINGABE


    ELSE BEGIN GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL END;


    IF NOCHMAL=FALSE THEN FELD:=SPALTE+(ZEILE-1)*3


    ELSE BEGIN FELD:=1;GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL END;


   UNTIL (NOT NOCHMAL) AND (F[FELD].ART=0);


   MACHZUG(F,B,FELD,MENSCH);


   TEXTCOLOR(M_COLOR);SETZEN(FELD,'O');NORMAL;


   GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL;GOTOXY(43,17);CLREOL;


  END;





 FUNCTION WERT(B:BEWERTUNG):INTEGER;


  VAR T,SP,COM: INTEGER;


  BEGIN


   SP:=0;COM:=0;INC(COUNT);


   FOR T:=1 TO 8 DO


    IF B[T].WERT=1 THEN COM:=COM+COSUM[B[T].ANZAHL]


    ELSE IF B[T].WERT=2 THEN SP:=SP+SPSUM[B[T].ANZAHL];


   WERT:=COM-SP


  END;





 PROCEDURE GENZUGLISTE(F:SPIELFELD;AMZUG:SPIELER;VAR N:BYTE;VAR ZL:SPIELZUGLISTE);


  VAR K:BYTE;


  BEGIN


   N:=0;


   FOR K:=1 TO 9 DO IF F[K].ART=0 THEN BEGIN INC(N);ZL[N]:=K END;


  END;





 FUNCTION BRETTVOLL(F:SPIELFELD):BOOLEAN;


  VAR K:BYTE;


  BEGIN


   BRETTVOLL:=FALSE;


   FOR K:=1 TO 9 DO


    IF F[K].ART=0 THEN EXIT;


   BRETTVOLL:=TRUE


  END;





 FUNCTION SCHLUSS(F:SPIELFELD;B:BEWERTUNG):BOOLEAN;


  VAR SIEG,K:WORD;


  BEGIN


   SIEG:=0;


   FOR K:=1 TO 8 DO


    BEGIN


     IF B[K].WERT=1 THEN


      BEGIN IF B[K].ANZAHL=3 THEN SIEG:=1 END


     ELSE


      IF B[K].WERT=2 THEN


       BEGIN IF B[K].ANZAHL=3 THEN SIEG:=2 END;


    END;


   SCHLUSS:=BRETTVOLL(F) OR (SIEG=1) OR (SIEG=2)


  END;





 PROCEDURE SUCHE_ZUG(VAR BESTZUG:SPIELZUG;VAR ALPHA_BETA:INTEGER;


                      B:BEWERTUNG;N:BYTE;F:SPIELFELD;AMZUG:SPIELER;ALPHA,BETA:INTEGER);


  VAR VIRFELD:SPIELFELD;


      ZUG:SPIELZUG;


      ZUGLISTE:SPIELZUGLISTE;


      VIRB:BEWERTUNG;


      M:INTEGER;


      I,ANZAHL,ZEILE,SPALTE:BYTE;


  BEGIN


   IF (N=0) OR SCHLUSS(F,B) THEN ALPHA_BETA:=WERT(B)


   ELSE


    BEGIN


     GENZUGLISTE(F,AMZUG,ANZAHL,ZUGLISTE);


     IF AMZUG=COMPUTER THEN


      BEGIN


       M:=ALPHA;I:=0;


       WHILE (I<ANZAHL) AND (M<BETA) DO


        BEGIN


         VIRFELD:=F;VIRB:=B;INC(I);ZUG:=ZUGLISTE[I];


         MACHZUG(VIRFELD,VIRB,ZUG,COMPUTER);


         SUCHE_ZUG(BESTZUG,ALPHA_BETA,VIRB,N-1,VIRFELD,MENSCH,M,BETA);


         IF (N=TIEFE) AND ANZEIGEN THEN


          BEGIN


           GOTOXY(5,18);CLREOL;FETT;


           ZEILE:=ZUG DIV 3+1;


           SPALTE:=ZUG MOD 3;


           IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(ZEILE) END;


           WRITE('Wert von Zug:',ZEILE:2,SPALTE:2,'=',ALPHA_BETA);


           NORMAL;GOTOXY(2,19);CLREOL;BLINK_LOW;


           WRITE(COUNT,' analysierte Stellungen');NORMAL;


           WARTE;


          END;


         IF M<ALPHA_BETA THEN


          BEGIN


           M:=ALPHA_BETA;


           IF N=TIEFE THEN


            BEGIN


             BESTZUG:=ZUG;


             IF ANZEIGEN THEN


              BEGIN


               GOTOXY(1,20);CLREOL;UNTERSTREICHEN;


               ZEILE:=BESTZUG DIV 3+1;


               SPALTE:=BESTZUG MOD 3;


               IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(ZEILE) END;


               WRITE('bisher bester Computerzug :',ZEILE:2,SPALTE:2);


               WRITE('=',ALPHA_BETA);NORMAL


              END


            END


          END


        END


      END


     ELSE


      BEGIN


       M:=BETA;I:=0;


       WHILE (I<ANZAHL) AND (M>ALPHA) DO


        BEGIN


         VIRFELD:=F;VIRB:=B;INC(I);ZUG:=ZUGLISTE[I];


         MACHZUG(VIRFELD,VIRB,ZUG,MENSCH);


         SUCHE_ZUG(BESTZUG,ALPHA_BETA,VIRB,N-1,VIRFELD,COMPUTER,ALPHA,M);


         IF M>ALPHA_BETA THEN M:=ALPHA_BETA


        END


      END;


     ALPHA_BETA:=M


    END


  END;





 PROCEDURE COMPUTERZUG;


  VAR MINMAXWERT:INTEGER;


      BESTZUG:SPIELZUG;


   BEGIN { OF COMPUTERZUG }


   GOTOXY(43,15); WRITE('Der Computer rechnet. Bitte warten!');CLREOL;


   COUNT:=0;


   SUCHE_ZUG(BESTZUG,MINMAXWERT,B,TIEFE,F,COMPUTER,MININT,MAXINT);


   MACHZUG(F,B,BESTZUG,COMPUTER);


   TEXTCOLOR(C_COLOR);SETZEN(BESTZUG,'X'); NORMAL;


  END;





 FUNCTION SPIELENDE:BOOLEAN;


  VAR K,SIEG,SIEGERREIHE:INTEGER;


  BEGIN


   SPIELENDE:=FALSE;SIEG:=0;


   FOR K:=1 TO 8 DO


    BEGIN


     IF B[K].WERT=1 THEN


      BEGIN IF B[K].ANZAHL=3 THEN BEGIN SIEG:=1;SIEGERREIHE:=K END END


     ELSE


      IF B[K].WERT=2 THEN


       BEGIN IF B[K].ANZAHL=3 THEN BEGIN SIEG:=2;SIEGERREIHE:=K END END


    END;


   IF SIEG=1 THEN


    BEGIN


     FETT;SPIELENDE:=TRUE;


     GOTOXY(51,18);WRITE('Ich habe gewonnen');


     FOR K:=1 TO 3 DO


      BEGIN BLINKEN;SETZEN(REIHE[SIEGERREIHE].FELDNR[K],'X');NORMAL END


    END;


   IF SIEG=2 THEN


    BEGIN


     FETT;SPIELENDE:=TRUE;


     GOTOXY(41,18);WRITE('Du hast ausnahmsweise gewonnen.');


     FOR K:=1 TO 3 DO


      BEGIN REVERSE;SETZEN(REIHE[SIEGERREIHE].FELDNR[K],'O');NORMAL END


    END


  END;





  BEGIN { of  Main}


   REPEAT


    CLRSCR;INIT;NEUES_SPIEL(DRAN);FILLCHAR(B,SIZEOF(B),0); { B=0 SETZEN }


    REPEAT


     IF DRAN=MENSCH THEN SPIELERZUG ELSE COMPUTERZUG;


     IF DRAN=MENSCH THEN DRAN:=COMPUTER ELSE DRAN:=MENSCH;


     INC(ZUEGE)


    UNTIL SPIELENDE OR (ZUEGE>9);


    IF ZUEGE>9 THEN


     BEGIN GOTOXY(41,18);WRITE('Remis') END;


    GOTOXY(46,23);NORMAL; WRITE(' Ein neues Spiel (J/N) ?  ');


    REPEAT CH:=UPCASE(READKEY) UNTIL CH IN['N','J'];


   UNTIL CH='N'


  END.

