{$R+,G+}

PROGRAM TICTACTOE_MIT_MINIMAX;

 USES CRT;

 TYPE SPIELFELD=ARRAY[1..9]OF BYTE;

      SPIELZUG=BYTE;

      SPIELER=(COMPUTER,MENSCH);

      SPIELZUGLISTE=ARRAY[1..9]OF SPIELZUG;



 CONST BELL=#7;ESC=#27;CODEFARBE=MAGENTA;M_COLOR=LIGHTBLUE;

       C_COLOR=LIGHTRED;MININT=-30000;MAXINT=30000;ENDE=9999;



 VAR STEIN:ARRAY[1..3]OF BYTE;

     F:SPIELFELD;

     B:ARRAY[1..8]OF BYTE;

     DRAN:SPIELER;

     TIEFE,SPALTE,REIHE,ZUG:BYTE;

     CH:CHAR;

     ANZEIGEN:BOOLEAN;

     COUNT:LONGINT;



 PROCEDURE BLINKEN;

  BEGIN TEXTCOLOR(LIGHTCYAN+BLINK) END;

 PROCEDURE REVERSE;

  BEGIN TEXTCOLOR(GREEN+BLINK) END;

 PROCEDURE FETT;

  BEGIN TEXTCOLOR(MAGENTA) END;

 PROCEDURE UNTERSTREICHEN;

  BEGIN TEXTCOLOR(BLUE) END;

 PROCEDURE BLINK_LOW;

  BEGIN TEXTCOLOR(CYAN) END;

 PROCEDURE NORMAL;

  BEGIN TEXTCOLOR(YELLOW) END;

 PROCEDURE WARTE;

  VAR C:CHAR;

  BEGIN

   GOTOXY(45,23);WRITE('Drücke die Leertaste');

   REPEAT C:=READKEY UNTIL C=' ';

   GOTOXY(45,23);CLREOL

  END;



 PROCEDURE NEUES_SPIEL(VAR AMZUG:SPIELER);

  VAR I: INTEGER;

  BEGIN

   FOR I:=1 TO 9 DO F[I]:=0;

   ZUG:=1;CLRSCR;NORMAL;

   GOTOXY(17,10);WRITE('Spielfeld');

   GOTOXY(15,11);WRITE('(((((((');

   GOTOXY(15,12);TEXTCOLOR(CODEFARBE);WRITE('1');NORMAL;WRITE('( ( ( ((');

   GOTOXY(15,13);TEXTCOLOR(CODEFARBE);WRITE('2');NORMAL;WRITE('( ( ( ((');

   GOTOXY(15,14);TEXTCOLOR(CODEFARBE);WRITE('3');NORMAL;WRITE('');

   GOTOXY(15,15);WRITE(' (((((((');

   GOTOXY(15,16);TEXTCOLOR(CODEFARBE);WRITE('   1  2  3');NORMAL;

   GOTOXY(40,1);WRITE('Tic Tac Toe  mit MiniMax');

   GOTOXY(35,3);WRITE('Ziel des Spiels ist, drei eigene Steine in');

   GOTOXY(35,4);WRITE('eine horizontale, diagonale oder vertikale');

   GOTOXY(35,5);WRITE('Reihe zu legen. ');

   GOTOXY(35,8);WRITE('Spielstärke (1-9) : ');

   REPEAT CH:=READKEY UNTIL CH IN['1'..'9'];WRITE(CH);

   TIEFE:=ORD(CH)-ORD('0');

   GOTOXY(35,17);WRITE('Wer soll beginnen (');TEXTCOLOR(WHITE);WRITE('M');

   NORMAL;WRITE(')ensch oder (');TEXTCOLOR(WHITE);WRITE('C');

   NORMAL;WRITE(')omputer ?');

   REPEAT CH:=UPCASE(READKEY) UNTIL CH IN[#13,'M','C'];

   WRITE(CH);DELAY(500);

   IF CH IN[#13,'M'] THEN AMZUG:=MENSCH ELSE AMZUG:=COMPUTER;

   GOTOXY(35,17);CLREOL;

   WRITE('Bewertung anzeigen (J/N) : ');

   REPEAT CH:=UPCASE(READKEY) UNTIL CH IN['J','N'];

   WRITE(CH);DELAY(500);

   IF CH='J' THEN ANZEIGEN:=TRUE ELSE ANZEIGEN:=FALSE;

   GOTOXY(35,17);CLREOL;

  END;



 PROCEDURE SETZEN(STEIN:CHAR);

  VAR I:INTEGER;

  BEGIN

   FOR I:=1 TO 5 DO

    BEGIN

     GOTOXY(SPALTE*3+15,11+REIHE);WRITE(' ');DELAY(70);

     GOTOXY(SPALTE*3+15,11+REIHE);WRITE(STEIN);DELAY(70)

    END

  END;



 PROCEDURE SPIELERZUG;

  VAR NOCHMAL:BOOLEAN;FELD:INTEGER;

  FUNCTION EINGABE:INTEGER;

   VAR CH:CHAR; I:INTEGER;

   BEGIN

    REPEAT

     WRITE(BELL);

     REPEAT CH:=READKEY UNTIL CH IN ['1'..'3',#27];

     IF CH=ESC THEN BEGIN NOCHMAL:=TRUE;EINGABE:=1 END

     ELSE I:=ORD(CH)-ORD('0');

    UNTIL CH IN ['1'..'3',#27];

    IF NOT NOCHMAL THEN BEGIN WRITE(CH);EINGABE:=I END

   END;



  BEGIN

   REPEAT

    NOCHMAL:=FALSE;

    GOTOXY(43,15);WRITE('Bitte gib deinen Zug ein : ');CLREOL;

    GOTOXY(52,19);WRITE(' Zeile  : ');

    IF NOT NOCHMAL THEN REIHE:=EINGABE

    ELSE BEGIN GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL END;

    GOTOXY(52,20);WRITE(' Spalte : ');

    IF NOT NOCHMAL THEN SPALTE:=EINGABE

    ELSE BEGIN GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL END;

    IF NOCHMAL=FALSE THEN FELD:=SPALTE+(REIHE-1)*3

    ELSE BEGIN FELD:=1;GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL END;

   UNTIL (NOT NOCHMAL) AND (F[FELD]=0);

   F[FELD]:=5;TEXTCOLOR(M_COLOR);SETZEN('O');NORMAL;

   GOTOXY(52,19);CLREOL;GOTOXY(52,20);CLREOL;GOTOXY(43,17);CLREOL;

  END;



 FUNCTION WERT(F:SPIELFELD):INTEGER;

  VAR T,C,M,SP,COM: INTEGER;

  BEGIN

   FOR T:=1 TO 3 DO

    BEGIN

     B[T]:=F[(T-1)*3+1]+F[(T-1)*3+2]+F[(T-1)*3+3];

     IF (B[T]=15) OR (B[T]=3) THEN

      BEGIN STEIN[1]:=(T-1)*3+1;STEIN[2]:=(T-1)*3+2;STEIN[3]:=(T-1)*3+3 END;

    END;

   FOR T:=1 TO 3 DO

    BEGIN

     B[T+3]:=F[T]+F[T+3]+F[T+6];

     IF (B[T+3]=15) OR (B[T+3]=3) THEN

      BEGIN STEIN[1]:=T;STEIN[2]:=T+3;STEIN[3]:=T+6 END;

    END;

   B[7]:=F[1]+F[5]+F[9];

   IF (B[7]=15) OR (B[7]=3) THEN

    BEGIN STEIN[1]:=1;STEIN[2]:=5;STEIN[3]:=9 END;

   B[8]:=F[3]+F[5]+F[7];

   IF (B[8]=15) OR (B[8]=3) THEN

    BEGIN STEIN[1]:=3;STEIN[2]:=5;STEIN[3]:=7 END;

   SP:=0;COM:=0;

   FOR T:=1 TO 8 DO

    BEGIN

     IF B[T]=1  THEN COM:=COM+2;

     IF B[T]=5  THEN SP:=SP+4;

     IF B[T]=2  THEN COM:=COM+4;

     IF B[T]=10 THEN SP:=SP+8;

     IF B[T]=3  THEN COM:=COM+15000;

     IF B[T]=15 THEN SP:=SP+20000;

    END;

   WERT:=COM-SP;INC(COUNT)

  END;



 PROCEDURE MACHZUG(VAR F:SPIELFELD;ZUG:SPIELZUG;AMZUG:SPIELER);

  BEGIN IF AMZUG=COMPUTER THEN F[ZUG]:=1 ELSE F[ZUG]:=5 END;



 PROCEDURE GENZUGLISTE(SF:SPIELFELD;VAR N:BYTE;VAR ZL:SPIELZUGLISTE);

  VAR K:BYTE;

  BEGIN

   N:=0;

   FOR K:=1 TO 9 DO

    IF SF[K]=0 THEN BEGIN INC(N);ZL[N]:=K END

  END;



 FUNCTION BRETTVOLL(F:SPIELFELD):BOOLEAN;

  VAR K:BYTE;

  BEGIN

   BRETTVOLL:=FALSE;

   FOR K:=1 TO 9 DO

    IF F[K]=0 THEN EXIT;

   BRETTVOLL:=TRUE

  END;



 FUNCTION SCHLUSS(F:SPIELFELD):BOOLEAN;

  BEGIN

   SCHLUSS:=BRETTVOLL(F) OR (ABS(WERT(F))>ENDE);

   DEC(COUNT);

  END;



 PROCEDURE SUCHE_ZUG(VAR BESTZUG:SPIELZUG;VAR  MINMAX:INTEGER;

                                                         N:BYTE;F:SPIELFELD;AMZUG:SPIELER);

  VAR VIRFELD:SPIELFELD;

      ZUG:SPIELZUG;

      ZUGLISTE:SPIELZUGLISTE;

      M:INTEGER;

      I,ANZAHL:BYTE;

  BEGIN

   IF (N=0) OR SCHLUSS(F) THEN MINMAX:=WERT(F)

   ELSE

    BEGIN

     GENZUGLISTE(F,ANZAHL,ZUGLISTE);

     IF AMZUG=COMPUTER THEN

      BEGIN

       M:=MININT;I:=0;

       WHILE I<ANZAHL DO

        BEGIN

         VIRFELD:=F;INC(I);ZUG:=ZUGLISTE[I];

         MACHZUG(VIRFELD,ZUG,COMPUTER);

         SUCHE_ZUG(BESTZUG,MINMAX,N-1,VIRFELD,MENSCH);

         IF (N=TIEFE) AND ANZEIGEN THEN

          BEGIN

           GOTOXY(5,18);CLREOL;FETT;

           REIHE:=ZUG DIV 3+1;

           SPALTE:=ZUG MOD 3;

           IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(REIHE) END;

           WRITE('Wert von Zug :',REIHE:2,SPALTE:2,'=',MINMAX);

           NORMAL;GOTOXY(2,19);CLREOL;BLINK_LOW;

           WRITE(COUNT,' analysierte Stellungen');NORMAL;

           WARTE;

          END;

         IF M<MINMAX THEN

          BEGIN

           M:=MINMAX;

           IF N=TIEFE THEN

            BEGIN

             BESTZUG:=ZUG;

             IF ANZEIGEN THEN

              BEGIN

               GOTOXY(1,20);CLREOL;UNTERSTREICHEN;

               REIHE:=BESTZUG DIV 3+1;

               SPALTE:=BESTZUG MOD 3;

               IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(REIHE) END;

               WRITE('bisher bester Computerzug :',REIHE:2,SPALTE:2);

               WRITE('=',MINMAX);NORMAL

              END

            END

          END

        END

      END

     ELSE

      BEGIN

       M:=MAXINT;I:=0;

       WHILE I<ANZAHL DO

        BEGIN

         VIRFELD:=F;INC(I);ZUG:=ZUGLISTE[I];

         MACHZUG(VIRFELD,ZUG,MENSCH);

         SUCHE_ZUG(BESTZUG,MINMAX,N-1,VIRFELD,COMPUTER);

         IF M>MINMAX THEN M:=MINMAX

        END

      END;

     MINMAX:=M

    END

  END;



 PROCEDURE COMPUTERZUG;

  VAR MINMAXWERT:INTEGER;

      BESTZUG:SPIELZUG;

  BEGIN { OF COMPUTERZUG }

   GOTOXY(43,15); WRITE('Der Computer rechnet. Bitte warten!');CLREOL;

   COUNT:=0;

   SUCHE_ZUG(BESTZUG,MINMAXWERT,TIEFE,F,COMPUTER);

   MACHZUG(F,BESTZUG,COMPUTER);

   REIHE:=BESTZUG DIV 3+1;SPALTE:=BESTZUG MOD 3;

   IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(REIHE) END;

   TEXTCOLOR(C_COLOR);SETZEN('X'); NORMAL;

   GOTOXY(37,11);CLREOL;WRITE('Computer setzt : ',REIHE:2,SPALTE:2);

  END;



 FUNCTION SPIELENDE:BOOLEAN;

  VAR COM,K:INTEGER;

  BEGIN

   SPIELENDE:=FALSE;COM:=WERT(F);DEC(COUNT);

   IF COM>ENDE THEN

    BEGIN

     TEXTCOLOR(15);SPIELENDE:=TRUE;

     GOTOXY(41,18);CLREOL;WRITE('Ich, der schlaue PC, habe gewonnen!');

     FOR K:=1 TO 3 DO

      BEGIN

       REIHE:=STEIN[K] DIV 3+1;SPALTE:=STEIN[K] MOD 3;

       IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(REIHE) END;

       BLINKEN;GOTOXY(SPALTE*3+15,11+REIHE);WRITE('X');NORMAL

      END

    END;

   IF COM<-ENDE THEN

    BEGIN

     TEXTCOLOR(15);SPIELENDE:=TRUE;

     GOTOXY(41,18);CLREOL;WRITE('Du hast ausnahmsweise gewonnen.');

     FOR K:=1 TO 3 DO

      BEGIN

       REIHE:=STEIN[K] DIV 3+1;SPALTE:=STEIN[K] MOD 3;

       IF SPALTE=0 THEN BEGIN SPALTE:=3;DEC(REIHE) END;

       REVERSE;GOTOXY(SPALTE*3+15,11+REIHE);WRITE('O');;NORMAL

      END

    END

  END;



  BEGIN { of Main}

   REPEAT

    NEUES_SPIEL(DRAN);

    REPEAT

     IF DRAN=MENSCH THEN SPIELERZUG ELSE COMPUTERZUG;

     IF DRAN=MENSCH THEN DRAN:=COMPUTER ELSE DRAN:=MENSCH;

     INC(ZUG)

    UNTIL SPIELENDE OR BRETTVOLL(F);

    IF NOT SPIELENDE THEN

     BEGIN GOTOXY(41,18);WRITE('Remis') END;

    GOTOXY(46,23);NORMAL; WRITE(' Ein neues Spiel (J/N) ?  ');

    REPEAT CH:=UPCASE(READKEY) UNTIL CH IN['N','J'];

   UNTIL CH='N'

  END.

